
Interfaces: Use


· In ABAP Objects, interfaces are implemented in addition to and independently of classes. Interfaces only describe the external point of contact of a class (protocols), they do not contain any implementation.

· Interfaces are usually defined by a user. The user describes in the interface which services (technical and semantic) it needs in order to carry out a task. The user never actually knows the providers of these services, but communicates with them through the interface. In this way the user is protected from actual implementations and can work in the same way with different classes/objects, as long as they provide the services required. This is known as polymorphism with interfaces.

Interfaces: Services


· In the above example, a client wants to use the same services of different server classes (lcl_carrier, lcl_hotel, and lcl_rental) in the same way. There are different implementations of the services in the server classes (polymorphism).

· The following are examples of shared services of the server classes in the above case: display_partner, check_availability, booking, or cancel_booking.

Interface: UML Example


· In UML, interfaces can represented in the same way as classes. However, they always have the stereotype «interface» above their name and can therefore be distinguished from classes. 

· The use of an interface is represented by a dotted line with a two-sided arrow from the user to the interface, the stereotype «uses» is optional. The fact that a class implements an interface is represented by a dotted line with a three-sided arrow from the class to the interface. The similarity to the representation of inheritance is intentional, because the interface can be seen as a generalization of the class implemented or the class can be seen as a specialization of the interface.

· In ABAP Objects, the same components can be defined in interfaces and in classes. This allows you to shift part of the public point of contact of a class into an interface, even though the class is already in use; users will not notice the difference as long as you use alias names (see appendix) for the components that are now in the interface.

· A class can implement any number of interfaces and an interface can be implemented by any number of classes. By implementing several interfaces, a class can simulate multiple inheritance. In the example, lcl_rental implements the interfaces lif_partners and lif_license.

· In ABAP Objects, the same components (attributes, methods, constants, types, alias names) can be defined in an interface in largely the same way as in classes. However, interfaces do not have component visibility levels.

· Interfaces are implemented in classes.

· The interface name is listed in the definition part of the class. Interfaces can only be implemented publicly and are therefore always in the PUBLIC SECTION. If you do not do this, you risk multiple implementations if a superclass and a subclass both implement the same interface privately.

· The operations defined in the interface are implemented as methods of a class. A check is carried out to ensure that all the methods defined in the interfaces are actually present in the implementation part of the class (for global interfaces, a missing or superfluous implementation of an interface method results in a warning).

· The attributes, events, constants, and types defined in the interface are automatically available to the class carrying out the implementation. 

· Interface components are addressed in the class carrying out the implementation by prefixing the interface name, followed by a tilde (the Interface Resolution Operator):  <interfacename>~<componentname>.
· Changes to an interface usually invalidate all the classes implementing it.

· Interfaces are the means of choice for describing external points of contact or protocols, without linking them to a type of implementation. An extra layer is introduced between the client and the server to protect the client explicitly from the server, thereby making the client independent.

· Interfaces enable you to work uniformly with different classes (providers/servers). In particular, they always ensure polymorphic behavior as they do not have their own implementation, but instead allow the providers to carry it out.

· The definition of an interface is always an abstraction: The user wants to handle various providers in the same way and must therefore abstract concrete implementations to a description of the services required to fulfill the task.

· You can also use interfaces to achieve multiple inheritance by defining the functionality to be inherited by a second class as an interface that the inheriting class then has to implement.

Working with Interface Components

Interface resolution operator


· You can access interface components using an object reference, whose class implements the interface.  Syntactically this is done with the interface resolution operator, just as with the method definitions in the implementation part of the class.

· This allows you to differentiate between components defined in the interface and components of the same name that are defined in the class itself. The reason for this is the shared namespace.

· To simplify accessing interface methods you can work with alias names.

· Alias names can only appear in the in the declaration part of a class or in the interface definition.

· Example for an alias in the interface:   ALIASES a1 FOR lif_interface~method1
An alias defined in this way can be directly addressed using  r_ref->a1.

Interface References: Narrowing Cast

· Interfaces are addressed using interface references. Interface references always refer to instances in the classes carrying out the implementation. Interface references always have both static and dynamic types.

· The assignment of an object reference to an interface reference is known as a narrowing cast since, as with inheritance, only a part of the object interface is visible once you have assigned the reference.

· With an interface reference, you can no longer address all components in the class carrying out the implementation, but only the components defined in the interface. These components are now addressed using the interface reference exclusively with their own short name.

· When an object reference is assigned to an interface reference, you must be able to convert the dynamic type of the interface reference to the static type of the object reference - that is, the class that was used to define the object reference must have implemented the interface of the interface-reference.

Using the Interface


· The travel agent is a user (client) of the created interface.  The travel agency keeps references to its business partners in an internal table. Th type of this internal table is REF TO lif_partners, that is the reference to the interface lif_partners.

· When it is called, the client's public method add_partner is transferred the business partner references. In the method, the references are inserted into the table partner_list. Therefore, the interface parameter of the add_partner method has the type REF TO lif_partners.

· The aim here is polymorphism: It needs to be possible to generically access the services of the interface later.

[image: image13.wmf]ã

SAP

AG 2002

METHOD display

_

agency

_

partners

.

DATA

: r_

partner

TYPE

REF

TO

lif

_

partners

.

LOOP AT partner

_

list INTO

r_

partner

.

r_

partner

-

>

display

_

partner

( ).

ENDLOOP

.

ENDMETHOD

.

METHOD

...~

display

_

partner

.   

display

_

hotel

_

attributes

( ). 

...

ENDMETHOD

.

METHOD ...~

display

_

partner

.   

display

_

rental

_

data

( ). 

...

ENDMETHOD

.

partner

_

list

(1)

lcl

_rental

(7)

lcl

_hotel

METHOD ...~

display

_

partner

.

display

_

attributes

( ). 

...

ENDMETHOD

.

(6) 

lcl

_carrier

Polymorphism and Interfaces


· Polymorphism can also be used for interfaces: you can use interface references to call methods that can have a different implementation depending on the object behind the reference. 

· The dynamic type, not the static type of the reference variable is used to search for the implementation of a method. In the above example, r_partner->display_partner( ) therefore uses the class of the instance that r_partner actually refers to to search for the implementation of display. The static type for r_partner, which is always REF TO lif_partners, is not used.

Interface References: Widening Cast


· The widening cast is, as with inheritance, the opposite of the narrowing cast: Here it is used to retrieve a class reference from an interface reference. Obviously it cannot be statically checked, since an interface can be implemented by more than one class. 

· An object reference cannot be assigned to an interface reference if it has itself not implemented the corresponding interface. It cannot be assigned even if a subclass has implemented the interface and the interface reference points to an object in this class.

· Assignments between interface references whose interfaces are not related to each other cannot be checked statically and must therefore be formulated using the cast operator ?=.
· For this type of assignment, a check must be carried out at runtime to see whether the class of the instance that the source reference points to also supports the interface that the target reference refers to. If this is the case, the cast is carried out, otherwise the exception that can be handled of the class CX_SY_MOVE_CAST_ERROR is raised.

· ABAP Objects contains a composition model for interfaces. A compound interface contains other interfaces as components (component interfaces) and summarizes the extension of these component interfaces. An elementary interface does not itself contain other interfaces.

· One interface can be used as a component interface in several compound interfaces.

· UML only deals with the specialization and generalization of interfaces. This relationship is represented by a dotted line with a three-sided arrow from the specialized to the generalized interface.

· Compound interfaces in ABAP Objects can always be seen as specializations of their component interfaces and represented as such in UML.

· In a compound interface, the components of the component interface keep their original names, that is <component-interfacename>~<componentname>; no more prefixes are added. In other words, all components in a compound interface are on the same level, and components inherited from component interfaces are marked with the usual interface prefix.

· This equality principle for compound interfaces also affects how they are implemented. The procedure is as follows: First, implement the elementary interfaces, then the additional methods from the compound interfaces. For multiple compound interfaces, the process is simply repeated. In the class carrying out the implementation, all components of all interfaces implemented are again on the same level.

· This means that interface components only ever exist once and are known by their original names <interfacename>~<componentname>. This is true both for compound interfaces and for the classes that implement them. 

	Unit: Interfaces

Topic: Defining Interfaces

	At the conclusion of these exercises, you will be able to:

· Define and implement interfaces

	Model solution:
SAPBC401_INTS_MAIN_A

SAPBC401_VEHD_F include program

SAPBC401_INTS_A include program


1
The classes lcl_rental, lcl_vehicle, and their subclasses are to be added to the existing UML diagram.
The classes lcl_rental, lcl_carrier, and later lcl_hotel will all store separate services and provide a possible client (to be defined later) using an interface lif_partners. 

1-1
Add the interface lif_partners to your UML diagram. display_partner is to be the only interface method. What are the roles of the classes lcl_rental, lcl_carrier, and lcl_hotel? Depict this in the UML diagram by adding text and arrows.
Fill in the UML diagram on the next page.

1-2
Copy the include program SAPBC401_VEHD_F to your own include program ZBC401_##_VEHICLE. As well as the classes lcl_rental, lcl_vehicle, lcl_truck, lcl_bus, the program also contains the interface lif_partners. 

1-3
Incorporate this include program into your main program. 
Add the types ty_fuel and ty_cargo (both TYPE P DECIMALS 2) to the main program.
Pay attention to the position of the interface definition; you may have to change the order of the include programs or move the interface into the other include program. Familiarize yourself with the interface definition.

1-4
Now implement the interface of your class lcl_carrier.

1-5
In the main program, create some instances for lcl_rental, lcl_vehicle, and so on by copying the relevant code from the template SAPBC401_VEHT_MAIN_A. Call the method display_attributes of the class lcl_rental.
Start your program. You should see an airline with your planes and a car rental company with various vehicles. 
(You will use the interface in the next exercise.)

UML-Template for Exercise 1-1

	Unit: Interfaces

Topic: Using Interfaces

	At the conclusion of these exercises, you will be able to:

· Define and implement interfaces

· Use polymorphism with interfaces

	Model solution:
SAPBC401_INTS_MAIN_B

SAPBC401_VEHD_G include program

SAPBC401_INTS_A include program


2
Now the class lcl_travel_agency (travel agency) is to be added to the existing UML diagram. It will use the interface to access the classes lcl_rental, lcl_carrier, lcl_hotel (not yet implemented). 

2-1
Add the class lcl_travel_agency to your UML diagram. What is the role of this class? Depict this in the UML diagram by adding text and arrows.
Fill in the UML diagram on the next page.

2-2
Copy the definition of the class lcl_travel_agency from the template SAPBC401_VEHT_B to your own include program. Then complete this new class.

2-2-1
The travel agency uses the public method add_partner to add business partners (that is car rental companies and airlines) to the internal table partner_list. 


2-2-2
What are the types of the internal table and the interface parameter of add_partner?


2-2-3
The travel agency will use the method display_agency_partners to display the data of its business partners.


2-3
Go back to your main program.

2-4
Create an instance of the class lcl_travel_agency.


2-4-1
Use the add_partner method to add the travel agency’s business partners to the internal table.

2-4-2
Use the display_agency_partners method to display the travel agency’s business partners.

2-4-3
At which point does polymorphism take place?
Follow this in the Debugger.

UML-Template for Exercise 2-1

	Unit:
Interfaces

Topic:
Defining Interfaces


*&---------------------------------------------------------------------*

*& Report  SAPBC401_INTS_MAIN_A                                        *

*&                                                                     *

*&---------------------------------------------------------------------*

*&  include interface lif_partners and include file with lcl_vehicle.. *

*&---------------------------------------------------------------------*

REPORT  sapbc401_ints_main_a.
types ty_fuel type p decimals 2.
types ty_cargo type p decimals 2.
INCLUDE <icon>.
INCLUDE sapbc401_vehd_f.
include sapbc401_ints_a.
DATA: r_plane TYPE REF TO lcl_airplane,
      r_cargo TYPE REF TO lcl_cargo_plane,
      r_passenger TYPE REF TO lcl_passenger_plane,
      r_carrier type ref to lcl_carrier,

r_agency type ref to lcl_travel_agency,
      r_rental type ref to lcl_rental,
      r_truck type ref to lcl_truck,
      r_bus type ref to lcl_bus.
START-OF-SELECTION.
*##############################

***** Create Carrier ********************************************
  create object r_carrier exporting im_name = 'Smile&Fly-Travel'.
***** Passenger Plane ********************************************
  CREATE OBJECT r_passenger EXPORTING
                         im_name = 'LH BERLIN'
                         im_planetype = '747-400'
                         im_seats = 345.
***** cargo Plane ************************************************
  CREATE OBJECT r_cargo EXPORTING
                         im_name = 'US HErcules'
                         im_planetype = '747-500'
                         im_cargo = 533.
***** insert planes into itab if client ***************************
  r_carrier->add_airplane( r_passenger ).
  r_carrier->add_airplane( r_cargo ).
***** show all airplanes inside carrier ***************************
  r_carrier->display_attributes( ).

******* create RENTAL *****************************************
  CREATE OBJECT r_rental EXPORTING im_name = 'HAPPY CAR RENTAL'.
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'MAN'
                                  im_cargo = 45.
  r_rental->add_vehicle( r_truck ).
******* create truck *****************************************
  CREATE OBJECT r_bus EXPORTING im_make = 'Mercedes'
                                im_passengers = 80.
  r_rental->add_vehicle( r_bus ).
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'VOLVO'
                                  im_cargo = 48.
  r_rental->add_vehicle( r_truck ).
*&---------------------------------------------------------------------*

*&  Include           SAPBC401_INTS_a                                  *
*&---------------------------------------------------------------------*

*------------------------------------------------------------------*

*       CLASS lcl_airplane DEFINITION                              *
*------------------------------------------------------------------*

...

*---------------------------------------------------------------------*

*       CLASS lcl_carrier DEFINITION
*---------------------------------------------------------------------*

*---------------------------------------------------------------------*

CLASS lcl_carrier DEFINITION.
  PUBLIC SECTION.
    "----------------------------------------

    INTERFACES lif_partners.
    METHODS: constructor IMPORTING im_name TYPE string,
             get_name RETURNING value(ex_name) TYPE string,
             add_airplane IMPORTING
                     im_plane TYPE REF TO lcl_airplane,
             display_airplanes,
             display_attributes.
  PRIVATE SECTION.
    "-----------------------------------

    DATA: name              TYPE string,
          airplane_list TYPE TABLE OF REF TO lcl_airplane.
ENDCLASS.                    "lcl_carrier DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_carrier IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_carrier IMPLEMENTATION.
  METHOD lif_partners~display_partner.
    display_airplanes( ).
  ENDMETHOD.                    "lif_partners~display_partner
  METHOD add_airplane.
    APPEND im_plane TO airplane_list.
  ENDMETHOD.                    "add_airplane
  METHOD display_attributes.
    WRITE: icon_flight AS ICON, name . ULINE. ULINE.
    display_airplanes( ).
  ENDMETHOD.                    "display_attributes
  METHOD display_airplanes.
    DATA: r_plane TYPE REF TO lcl_airplane.
    LOOP AT airplane_list INTO r_plane.
      r_plane->display_attributes( ).
    ENDLOOP.
  ENDMETHOD.                    "display_airplanes
  METHOD constructor.
    name = im_name.
  ENDMETHOD.                    "constructor
  METHOD get_name.
    ex_name = name.
  ENDMETHOD.                    "get_name
ENDCLASS.                    "lcl_carrier IMPLEMENTATION
Solutions

	Unit:
Interfaces

Topic:
Using Interfaces


*&---------------------------------------------------------------------*

*& Report  SAPBC401_INTS_MAIN_B                                        *

*&                                                                     *

*&---------------------------------------------------------------------*

*&  include interface lif_partners and include file with lcl_vehicle.. *

*&  the client (travel_agency) uses the interface...                   *

*&---------------------------------------------------------------------*

REPORT  sapbc401_ints_main_b.
types ty_fuel type p decimals 2.
types ty_cargo type p decimals 2.
INCLUDE <icon>.
INCLUDE sapbc401_vehd_g.
include sapbc401_ints_a.
DATA: r_plane TYPE REF TO lcl_airplane,
      r_cargo TYPE REF TO lcl_cargo_plane,
      r_passenger TYPE REF TO lcl_passenger_plane,
      r_carrier type ref to lcl_carrier,
      r_agency type ref to lcl_travel_agency,
      r_rental type ref to lcl_rental,
      r_truck type ref to lcl_truck,
      r_bus type ref to lcl_bus.
START-OF-SELECTION.
*##############################

***** Create TRAVEL_AGENCY **************************************
    CREATE OBJECT r_agency EXPORTING im_name = 'Fly&Smile Travel'.
***** Create CARRIER ********************************************
  create object r_carrier exporting im_name = 'Smile&Fly-Travel'.
***** Passenger Plane ********************************************
  CREATE OBJECT r_passenger EXPORTING
                         im_name = 'LH BERLIN'
                         im_planetype = '747-400'
                         im_seats = 345.
***** cargo Plane ************************************************
  CREATE OBJECT r_cargo EXPORTING
                         im_name = 'US HErcules'
                         im_planetype = '747-500'
                         im_cargo = 533.
***** insert planes into itab if client ***************************
  r_carrier->add_airplane( r_passenger ).
  r_carrier->add_airplane( r_cargo ).
***** insert business-parnter of agency into partner_list***********
  r_agency->add_partner( r_carrier ).
******* create RENTAL *****************************************
  CREATE OBJECT r_rental EXPORTING im_name = 'HAPPY CAR RENTAL'.
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'MAN'
                                  im_cargo = 45.
  r_rental->add_vehicle( r_truck ).
******* create truck *****************************************
  CREATE OBJECT r_bus EXPORTING im_make = 'Mercedes'
                                im_passengers = 80.
  r_rental->add_vehicle( r_bus ).
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'VOLVO'
                                  im_cargo = 48.
  r_rental->add_vehicle( r_truck ).
***** insert business-partner of agency into partner_list***********
  r_agency->add_partner( r_rental ).
******* show attributes of all partners of travel_agency ******
  r_agency->display_agency_partners( ).
*&---------------------------------------------------------------------*

*&  Include           SAPBC401_VEHD_g                                  *
*&---------------------------------------------------------------------*

*---------------------------------------------------------------------*

* define client lcl_travel_agency
* it will use the interface lif_partners
*---------------------------------------------------------------------*

INTERFACE lif_partners.
  METHODS display_partner.
ENDINTERFACE.                    "lif_partners
*---------------------------------------------------------------------*

*       CLASS lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_vehicle DEFINITION.
  PUBLIC SECTION.
    "-------------------

    METHODS: get_average_fuel IMPORTING im_distance TYPE s_distance
                                        im_fuel TYPE ty_fuel
                              RETURNING value(re_avgfuel) TYPE ty_fuel.
    METHODS       constructor IMPORTING im_make TYPE string.
    METHODS       display_attributes.
    METHODS       set_make IMPORTING im_make TYPE string.
    METHODS       get_make EXPORTING ex_make TYPE string.
    CLASS-METHODS get_count EXPORTING re_count TYPE i.
  PRIVATE SECTION.
    "-------------------

    DATA: make   TYPE string.
    METHODS      init_make.
    CLASS-DATA:  n_o_vehicles TYPE i.
ENDCLASS.                    "lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_vehicle IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_vehicle IMPLEMENTATION.
  METHOD get_average_fuel.
    re_avgfuel = im_distance / im_fuel.
  ENDMETHOD.                    "get_average_fuel
  METHOD constructor.
    make = im_make.
    ADD 1 TO n_o_vehicles.
  ENDMETHOD.                    "constructor
  METHOD set_make.
    IF im_make IS INITIAL.
      init_make( ).   " me->init_make( ). also possible
    ELSE.
      make = im_make.
    ENDIF.
  ENDMETHOD.                    "set_make
  METHOD init_make.
    make = 'default make'.
  ENDMETHOD.                    "init_make
  METHOD get_make.
    ex_make = make.
  ENDMETHOD.                    "get_make
  METHOD display_attributes.
    WRITE: make.
  ENDMETHOD.                    "display_attributes
  METHOD get_count.
    re_count = n_o_vehicles.
  ENDMETHOD.                    "get_count
ENDCLASS.                    "lcl_vehicle IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_truck DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_truck DEFINITION INHERITING FROM lcl_vehicle.
  PUBLIC SECTION.
    "-------------------

    METHODS:      constructor IMPORTING im_make TYPE string
                                        im_cargo TYPE ty_cargo.
    METHODS       display_attributes REDEFINITION.
    METHODS       get_cargo RETURNING value(re_cargo) TYPE ty_cargo.
  PRIVATE SECTION.
    "-------------------

    DATA: max_cargo TYPE ty_cargo.
ENDCLASS.                    "lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_truck IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_truck IMPLEMENTATION.
  METHOD constructor.
    super->constructor( im_make ).
    max_cargo = im_cargo.
  ENDMETHOD.                    "constructor
  METHOD display_attributes.
    WRITE: / icon_ws_truck AS ICON.
    super->display_attributes( ).
    WRITE: 20 ' Cargo = ', max_cargo.
    ULINE.
  ENDMETHOD.                    "display_attributes
  METHOD get_cargo.
    re_cargo = max_cargo.
  ENDMETHOD.                    "get_cargo
ENDCLASS.                    "lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_bus DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_bus DEFINITION INHERITING FROM lcl_vehicle.
  PUBLIC SECTION.
    "-------------------

    METHODS:      constructor IMPORTING im_make TYPE string
                                        im_passengers TYPE i.
    METHODS       display_attributes REDEFINITION.
  PRIVATE SECTION.
    "-------------------

    DATA: max_passengers TYPE i.
ENDCLASS.                    "lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_bus IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_bus IMPLEMENTATION.
  METHOD constructor.
    super->constructor( im_make ).
    max_passengers = im_passengers.
  ENDMETHOD.                    "constructor
  METHOD display_attributes.
    WRITE: / icon_transportation_mode AS ICON.
    super->display_attributes( ).
    WRITE:  20 ' Passengers = ', max_passengers.
    ULINE.
  ENDMETHOD.                    "display_attributes
ENDCLASS.                    "lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_rental DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_rental DEFINITION.
  PUBLIC SECTION.
    "-------------------

    METHODS:      constructor IMPORTING im_name TYPE string.
    METHODS       add_vehicle IMPORTING im_vehicle
                                  TYPE REF TO lcl_vehicle.
    METHODS       display_attributes.
    INTERFACES: lif_partners.
  PRIVATE SECTION.
    "-------------------

    DATA: name TYPE string,
          vehicle_list TYPE TABLE OF REF TO lcl_vehicle.
ENDCLASS.                    "lcl_rental DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_rental IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_rental IMPLEMENTATION.
  METHOD lif_partners~display_partner.
    display_attributes( ).
  ENDMETHOD.                    "lif_partners~display_partner
  METHOD  constructor.
    name = im_name.
  ENDMETHOD.                    "constructor
  METHOD  add_vehicle.
    APPEND im_vehicle TO vehicle_list.
  ENDMETHOD.                    "add_vehicle
  METHOD  display_attributes.
    DATA: r_vehicle TYPE REF TO lcl_vehicle.
    WRITE: /  icon_transport_proposal AS ICON, name.
    WRITE:  ' Here comes the vehicle list: '. ULINE. ULINE.
    LOOP AT vehicle_list INTO r_vehicle.
      r_vehicle->display_attributes( ).
    ENDLOOP.
  ENDMETHOD.                    "display_attributes
ENDCLASS.                    "lcl_rental IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_travel_agency DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_travel_agency DEFINITION.
  PUBLIC SECTION.
    "-------------------

    METHODS:      constructor IMPORTING im_name TYPE string.
    METHODS       add_partner IMPORTING im_partner
                                  TYPE REF TO lif_partners.
    METHODS       display_agency_partners.
  PRIVATE SECTION.
    "-------------------

    DATA: name TYPE string,
          partner_list TYPE TABLE OF REF TO lif_partners.
ENDCLASS.                    "lcl_travel_agency DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_travel_agency IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_travel_agency IMPLEMENTATION.
  METHOD display_agency_partners.
    DATA: r_partner TYPE REF TO lif_partners.
    WRITE: icon_dependents AS ICON, name.
    WRITE:  ' Here are the partners of the travel agency: '.ULINE.ULINE.
    LOOP AT partner_list INTO r_partner.
      r_partner->display_partner( ).
    ENDLOOP.
ENDMETHOD.                    "display_agency_partners
METHOD  constructor.
  name = im_name.
ENDMETHOD.                    "constructor
METHOD  add_partner.
  APPEND im_partner TO partner_list.
ENDMETHOD.                    "add_partner
ENDCLASS.                    "lcl_travel_agency IMPLEMENTATION

